
 

 

Judd Communications Depth Sensor Digital Output Option 
 
Judd Communications Depth Sensors equipped with the digital output option can output an ASCII sentence. 
After each measurement, 4 parameters are output serially as a line of ASCII characters at 1200 baud (8N1). 
To monitor the digital output, connect the depth sensor to an RS-232 serial port as follows: 
 
Depth Sensor wire RS-232 serial port 25 pin (DTE) 9 pin (DTE) 
White Receive 3 2 
Black Ground 7 5 
 
Run your communications software on your PC and configure it for the COM port that the depth sensor is 
connected to, set the baud rate to 1200, and set the data length to 8 bits, no parity, and 1 stop bit.  
 
Apply power to the depth sensor, connect the red(+) and black(-) wire to a +12 Volt power source. Connect 
the green wire to the +12 Volt source to activate the sensor. To repeat the measurement, disconnect the 
green wire and then reconnect. 
 
A line of text should be displayed by your communications software on your PC. The line consists of 4 
parameters and should look something like this: 
 
595 1674 2904 0 
 
Each parameter is separated by a TAB character (ASCII value 09), and the line is terminated by a Carriage 
Return / Line Feed combination (ASCII values 13,10). 
 
The 4 parameters are defined as follows: 
 

1. Air temperature in degrees Kelvin * 2  (595 = 297.5 * 2) 

2. The time in tens of microseconds that it took to for the ultrasonic pulse echo to return. 

3. The distance measurement in millimeters, calculated by using the first two parameters.  

Speed of sound, 346.27 meters/second = 331.4 + (.607 * (Temp K -273)) 

Distance to Target, ~ 2.904 meters = (.01674 seconds / 2) * 346.27 meters/second 

4. The number of tries the retry/error checking algorithm was used beyond the first pass. 

 

Programming Example for use with a Campbell CR10 or CR 10X datalogger 

;{CR10} 

;Judd Communications Depth Sensor with 

;Digital Output option, programming example 

 

;Wiring 

;Red        +12V 

;Black      G 

;Clear      G 

;Green      C1 

;White      C3 

;Brown      Not used 

 



 

 

*Table 1 Program 

  01: 3         Execution Interval (seconds) 

 

1:  Batt Voltage (P10) 

 1: 1        Loc [ Battery   ] 

 

;The depth sensor sampling routine works best as a  

;subroutine. The following command will call  

;subroutine #1 once an hour. 

2:  If time is (P92) 

 1: 0        Minutes (Seconds --) into a 

 2: 60       Interval (same units as above) 

 3: 01       Call Subroutine 1 

 

;The following command will allow the user to activate  

;the depth sensor by using the "F1" Flag while  

;monitoring the input locations using by Graph TERM  

;program or the Connect button in PC208W. 

3:  If Flag/Port (P91) 

 1: 11       Do if Flag 1 is High 

 2: 01       Call Subroutine 1 

 

;------------------------------------------------ 

;Log sensor output once an hour 

 

4:  If time is (P92) 

 1: 0        Minutes (Seconds --) into a 

 2: 60       Interval (same units as above) 

 3: 10       Set Output Flag High 

 

5:  Set Active Storage Area (P80) 

 1: 1        Final Storage Area 1 

 2: 101      Array ID or Loc [ _________ ] 

 

6:  Real Time (P77) 

 1: 1110     Year,Day,Hour/Minute 

 

7:  Sample (P70) 

 1: 1        Reps 

 2: 1        Loc [ Battery   ] 

 

8:  Sample (P70) 

 1: 1        Reps 

 2: 2        Loc [ DSTempF   ] 

 

9:  Sample (P70) 

 1: 1        Reps 

 2: 3        Loc [ DSTime    ] 

 

10:  Sample (P70) 

 1: 1        Reps 

 2: 4        Loc [ DSDepth   ] 

 

11:  Sample (P70) 

 1: 1        Reps 

 2: 5        Loc [ DSRetries ] 

 

;------------------------------------------- 

;Table 2 not used 

*Table 2 Program 

  02: 0.0000    Execution Interval (seconds) 

 

;------------------------------------------- 

*Table 3 Subroutines 

1:  Beginning of Subroutine (P85) 

 1: 1        Subroutine 1 

 

;Turn Depth Sensor ON using Control port 1 

2:  Do (P86) 

 1: 41       Set Port 1 High 

 

;Wait 2.75 seconds for ASCII sentence from 
sensor 

;Data is input to Control port 3. Control port 2 

;is used by the P15 instruction as a DTR line. 

;Nothing is connected to Control port 2 

3:  Port Serial I/O (Special) (P15) 

 1: 1        Reps 

 2: 1        ASCII/RS-232, 1200 Baud 

 3: 1        delay (0.01 seconds) 

 4: 2        First Control Port 

 5: 6        Output Loc [ Dummy     ] 

 6: 0        No. of Locs to Send 

 7: 10       Termination Character 

 8: 26       Maximum Characters 

 9: 275      CTS/Input Wait 

10: 2        Loc [ DSTempF   ] 

11: 1.0      Mult 

12: 0.0      Offset 

 

 

;Turn Depth Sensor OFF 

4:  Do (P86) 

 1: 51       Set Port 1 Low 

 

;Convert Air Temp from Kelvin x 2, to Farenheit 

;From K to C 

5:  Z=X*F (P37) 

 1: 2        X Loc [ DSTempF   ] 

 2: .5       F 

 3: 2        Z Loc [ DSTempF   ] 

 

6:  Z=X+F (P34) 

 1: 2        X Loc [ DSTempF   ] 

 2: -273     F 

 3: 2        Z Loc [ DSTempF   ] 

 

;From C to F 

7:  Z=X*F (P37) 

 1: 2        X Loc [ DSTempF   ] 

 2: 1.8      F 

 3: 2        Z Loc [ DSTempF   ] 

 

8:  Z=X+F (P34) 

 1: 2        X Loc [ DSTempF   ] 

 2: 32       F 

 3: 2        Z Loc [ DSTempF   ] 

 

;Convert Distance measured from millimeters to 

;inches, and convert to a negative number 

9:  Z=X*F (P37) 

 1: 4        X Loc [ DSDepth   ] 

 2: -.03937  F 

 3: 4        Z Loc [ DSDepth   ] 

;Convert distance measured to depth  

; Depth = Distance to ground - Distance measured 

;10:  Z=X+F (P34) 

; 1: 4        X Loc [ DSDepth   ] 

; 2: 0        F <<<<<<<<<<<<<<<<<< DISTANCE TO 
GROUND GOES HERE!!! 

; 3: 4        Z Loc [ DSDepth   ] 

10:  End (P95) 

 

End Program 

